pragma solidity ^0.4.16;

interface tokenRecipient { function receiveApproval(address _from, uint256 _value, address _token, bytes _extraData) public; }

contract DailyCoinToken {
    // Public variables of the token
    string public name;
    string public symbol;
    uint8 public decimals = 8;
    // 18 decimals is the strongly suggested default, avoid changing it
    uint256 public totalSupply;

    // This creates an array with all balances
    mapping (address => uint256) public balanceOf;
    mapping (address => mapping (address => uint256)) public allowance;

    // This generates a public event on the blockchain that will notify clients
    event Transfer(address indexed from, address indexed to, uint256 value);

    // This notifies clients about the amount burnt
    event Burn(address indexed from, uint256 value);

    /**
     * Constrctor function
     *
     * Initializes contract with initial supply tokens to the creator of the contract
     */
    function DailyCoinToken(
    ) public {
        totalSupply = 300000000 * 10 ** uint256(decimals);  // Update total supply with the decimal amount
        balanceOf[msg.sender] = totalSupply;                // Give the creator all initial tokens
        name = "Daily Coin";                                   // Set the name for display purposes
        symbol = "DLC";                               // Set the symbol for display purposes
    }

    /**
     * Internal transfer, only can be called by this contract
     */
    function _transfer(address _from, address _to, uint _value) internal {
        // Prevent transfer to 0x0 address. Use burn() instead
        require(_to != 0x0);
        // Check if the sender has enough
        require(balanceOf[_from] >= _value);
        // Check for overflows
        require(balanceOf[_to] + _value > balanceOf[_to]);
        // Save this for an assertion in the future
        uint previousBalances = balanceOf[_from] + balanceOf[_to];
        // Subtract from the sender
        balanceOf[_from] -= _value;
        // Add the same to the recipient
        balanceOf[_to] += _value;
        Transfer(_from, _to, _value);
        // Asserts are used to use static analysis to find bugs in your code. They should never fail
        assert(balanceOf[_from] + balanceOf[_to] == previousBalances);
    }

    /**
     * Transfer tokens
     *
     * Send `_value` tokens to `_to` from your account
     *
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transfer(address _to, uint256 _value) public {
        _transfer(msg.sender, _to, _value);
    }

    /**
     * Transfer tokens from other address
     *
     * Send `_value` tokens to `_to` in behalf of `_from`
     *
     * @param _from The address of the sender
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transferFrom(address _from, address _to, uint256 _value) public returns (bool success) {
        require(_value <= allowance[_from][msg.sender]);     // Check allowance
        allowance[_from][msg.sender] -= _value;
        _transfer(_from, _to, _value);
        return true;
    }

    /**
     * Set allowance for other address
     *
     * Allows `_spender` to spend no more than `_value` tokens in your behalf
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     */
    function approve(address _spender, uint256 _value) public
        returns (bool success) {
        allowance[msg.sender][_spender] = _value;
        return true;
    }

    /**
     * Set allowance for other address and notify
     *
     * Allows `_spender` to spend no more than `_value` tokens in your behalf, and then ping the contract about it
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     * @param _extraData some extra information to send to the approved contract
     */
    function approveAndCall(address _spender, uint256 _value, bytes _extraData)
        public
        returns (bool success) {
        tokenRecipient spender = tokenRecipient(_spender);
        if (approve(_spender, _value)) {
            spender.receiveApproval(msg.sender, _value, this, _extraData);
            return true;
        }
    }

    /**
     * Destroy tokens
     *
     * Remove `_value` tokens from the system irreversibly
     *
     * @param _value the amount of money to burn
     */
    function burn(uint256 _value) public returns (bool success) {
        require(balanceOf[msg.sender] >= _value);   // Check if the sender has enough
        balanceOf[msg.sender] -= _value;            // Subtract from the sender
        totalSupply -= _value;                      // Updates totalSupply
        Burn(msg.sender, _value);
        return true;
    }

    /**
     * Destroy tokens from other account
     *
     * Remove `_value` tokens from the system irreversibly on behalf of `_from`.
     *
     * @param _from the address of the sender
     * @param _value the amount of money to burn
     */
    function burnFrom(address _from, uint256 _value) public returns (bool success) {
        require(balanceOf[_from] >= _value);                // Check if the targeted balance is enough
        require(_value <= allowance[_from][msg.sender]);    // Check allowance
        balanceOf[_from] -= _value;                         // Subtract from the targeted balance
        allowance[_from][msg.sender] -= _value;             // Subtract from the sender's allowance
        totalSupply -= _value;                              // Update totalSupply
        Burn(_from, _value);
        return true;
    }
}

///////////////////////////CROWDSALE///////////////////////////////
contract DailycoinCrowdsale {
    uint256 public amountRaised = 0;
	uint256 public tokensSold = 0;
    uint256 public totalToSale = 150 * (10**6) * (10**8);
	bool crowdsaleClosed = false;
	
    uint public deadline;
    address public beneficiary;
    DailyCoinToken public tokenReward;

    event SaleEnded(address recipient, uint256 totalAmountRaised);
    event FundTransfer(address backer, uint256 amount, uint256 numOfTokens);

    /**
     * Constrctor function
     *
     * Setup the owner
     */
    function DailycoinCrowdsale() public {
        beneficiary = 0x17Cb4341eF4d9132f9c86b335f6Dd6010F6AeA9a;
        tokenReward = DailyCoinToken(0xaA33983Acfc48bE1D76e0f8Fe377FFe956ad84AD);
        deadline = 1512997200 + 45 days; // start on 2017-Dec-11 8.00PM, last 45 days
    }

    /**
     * Fallback function
     *
     * The function without name is the default function that is called whenever anyone sends funds to a contract
     */
    function () payable public {
		require(!crowdsaleClosed);
        uint256 amount = msg.value;
		uint256 numOfTokens = getNumTokens(amount);
        amountRaised += amount;
		tokensSold += numOfTokens;
        tokenReward.transfer(msg.sender, numOfTokens);
        FundTransfer(msg.sender, amount, numOfTokens);
    }
	
	function getNumTokens(uint256 _value) internal returns (uint256 numTokens) {
		uint256 multiple = 5000;
        if (_value >= 10 * 10**18) {
            if (now <= deadline - 35 days) { // first 10 days
				multiple = multiple * 130 / 100;
			} else if (now <= deadline - 20 days) { // next 15 days
				multiple = multiple * 120 / 100;
			} else { // 20 last days
				multiple = multiple * 115 / 100;
			}
        } else {
			if (now <= deadline - 35 days) { // first 10 days
				multiple = multiple * 120 / 100;
			} else if (now <= deadline - 20 days) {  // next 15 days
				multiple = multiple * 110 / 100;
			} else { // 20 last days
				multiple = multiple * 105 / 100;
			}
		}
		return multiple * 10**8 * _value / 10**18;
	}

    modifier afterDeadline() { if (now >= deadline) _; }

    /**
     * Withdraw the funds
     *
     * Checks to see if goal or time limit has been reached, and if so, give raised value to beneficiary and burn left tokens in crowdsale.
     */
    function endFunding() afterDeadline public {
		require(beneficiary == msg.sender);
		require(!crowdsaleClosed);
		if (beneficiary.send(amountRaised)) {
			if (totalToSale > tokensSold) {
				tokenReward.burn(totalToSale - tokensSold);
			}
			crowdsaleClosed = true;
			SaleEnded(beneficiary, amountRaised);
		}
    }
	
	function withdraw(uint256 amount) afterDeadline public {
		require(beneficiary == msg.sender);
		amount = amount * 1 ether;
		beneficiary.transfer(amount);
    }
}